Simpson's Rule

Simpson's Rule takes advantage of the fact that most functions are smooth curves, not straight line segments. Simpson's rule fits a *parabola* through the top three corners of each pair of strips and sums up the area under all of the parabolas. In Fig. 6.16 we have shown one pair of strips and the parabola that fits this pair. Our first task will be to find the correct values of a, b and c in order that the parabola $y = a\,x^2 + b\,x + c$ goes through the top three corners. Once we have found them then it is straightforward to find the area under the parabola. It is

$$\int_{-\Delta x}^{+\Delta x} (a\,x^2 + b\,x + c)\,dx \;=\; \left[\tfrac{1}{3} a\,x^3 + \tfrac{1}{2} b\,x^2 + c\,x\right]_{x=-\Delta x}^{x=+\Delta x} \;=\; \tfrac{2}{3} a\,\Delta x^3 + 2c\,\Delta x\,. \qquad (6.27)$$

Now let's turn to the task of finding the values of a, b and c. As shown in Fig. 6.16, the three points that the parabola must go through have coordinates $(-\Delta x, y_L)$, $(0, y_M)$ and $(+\Delta x, y_R)$. Substituting these into the equation $y = a\,x^2 + b\,x + c$ gives three equations in the three unknowns a, b and c:

$$\begin{cases} y_R = a\,\Delta x^2 + b\,\Delta x + c \\ y_M = a\,0^2 + b\,0 + c \\ y_L = a\,(-\Delta x)^2 + b(-\Delta x) + c \end{cases} \qquad (6.28)$$

The second equation immediately gives $c = y_M$. Substituting this into the first and third equations gives the following pair of equations in the unknowns a and b:

$$\begin{cases} \Delta x^2 \cdot a + \Delta x \cdot b = y_R - y_M \\ \Delta x^2 \cdot a - \Delta x \cdot b = y_L - y_M \end{cases} \qquad (6.29)$$

Adding the equations, we get

$$a \;=\; \frac{y_L + y_R - 2y_M}{2\,\Delta x^2}\,. \qquad (6.30)$$

Substituting the values found for a and c into the formula for the area under the parabola given on the previous page, we get the formula shown on the next page for the area under the parabola.



Figure 6.16 Simpson's rule breaks the integration region into a large number of *pairs of strips*. It then fits a parabola through the top three corners of each pair. This figure shows one such pair.

$$\text{area under parabola} \;=\; \frac{2}{3}a\,\Delta x^3 + 2c\,\Delta x$$

$$= \frac{\Delta x}{3}(y_L + y_R - 2y_M) + 2y_M\,\Delta x$$

$$= \frac{\Delta x}{3}(1y_L + 4y_M + 1y_R) \qquad (6.31)$$

Notice that the coefficients of y_L, y_M and y_R inside the brackets are 1, 4 and 1. Notice also that the area does not depend on where the pair of strips is located. We located them at the origin only to make the task of finding *a*, *b* and *c* easier. Now suppose that we have *three* strip pairs. Then the area under three parabolas is

$$\text{area under 3 parabolas} \;=\; \frac{\Delta x}{3}(y_0 + 4y_1 + y_2) + \frac{\Delta x}{3}(y_2 + 4y_3 + y_4) + \frac{\Delta x}{3}(y_4 + 4y_5 + y_6)\,,$$

or rearranging,

$$\text{area under 3 parabolas} \;=\; \frac{\Delta x}{3}\left(1y_0 + 4y_1 + 2y_2 + 4y_3 + 2y_4 + 4y_5 + 1y_6\right). \qquad (6.32)$$

Notice the pattern of the coefficients of the *y*'s inside the brackets; it starts with 1 4 and ends with 4 1, and in the interior it alternates 4, 2, 4, 2, ... This pattern generalizes to any number of <u>pairs</u> of strips.

$$\text{area of } n \text{ strips} \;=\; \frac{\Delta x}{3}\left(y_0 + 4y_1 + 2y_2 + 4y_3 + 2y_4 + \cdots + 4y_{n-1} + y_n\right). \qquad (6.33)$$

Notice that in this formula *n* must be an even number. (This is because the region between $x=a$ and $x=b$ must be divided in such a way that all strips occur in pairs.) As usual the first and last *y*'s represent the value of the function at the endpoints $x=a$ and at $x=b$.

Here is Visual Basic code that implements Simpson's Rule. To use it see the instructions on page 118.

```
Public Function Simpson (A, B, N)
   Dim X As Double, Dx As Double, Sum As Double, I As Integer
   'this function integrates a function called Func(x) between x=A and x=B
   'using Simpson's rule with N strips.

   N = (N \ 2) * 2                'Make sure N is even. The \ operator does integer division.
   Dx = (B - A) / N               'Dx is the width of a strip
   Sum = 0

   For I = 0 To N                 'the main loop
      X = A + I * Dx
      If I = 0 Or I = N Then      'I for one of the endpoints
         Sum = Sum + Func(X)
      ElseIf I Mod 2 = 1 Then     'I is odd
         Sum = Sum + 4 * Func(X)
      Else                        'I is even
         Sum = Sum + 2 * Func(X)
      End If
   Next I

   Simpson = Sum * Dx / 3
End Function
```

```
Public Function Func(X)            'Func is the integrand
   Func = X ^ 3 - X                'change this line for other integrands
End Function
```

As before the program is set up to calculate the definite integral

$$\int_2^5 (x^3 - x)\,dx \;.$$

A, *B*, *Func(X)*, *N*, *Dx* and *Sum* are defined the same as in the code for the Trapezoidal Rule. By the end of the “For” loop *Sum* holds the value $y_0 + 4y_1 + 2y_2 + \cdots + 4y_{n-1} + y_n$. The “If” statement inside the “For” loop guarantees that the 4’s and 2’s go in the right places. After the loop is completed *Sum* is multiplied by *Dx* / 3 and this value is copied to *Simpson*, which is then returned.

Running the Simpson’s Rule program with N=2 (the absolute minimum number of strips) already gives exactly the known value of 141.750. We see that the accuracy is <u>much better</u> than that of the trapezoidal rule.
